

# Dependent lenses

Nils Anders Danielsson

University of Gothenburg

nad@cse.gu.se

## Abstract

Very well-behaved lenses provide a convenient mechanism for defining setters and getters for nested records (among other things). However, they do not work very well for dependent records, in which one field’s type can depend on the value of a previous field.

This paper discusses dependent lenses, a generalisation of ordinary lenses with better support for dependent records. It is shown that a certain notion of composition cannot be defined for such lenses (under certain assumptions), but that another, different notion of composition can be defined.

**Keywords** Lenses, Dependent types, Agda

## 1. Introduction

Very well-behaved lenses (Foster et al. 2005)—in this paper I only consider the total variant—are used to make programs that work with nested record structures more elegant. Consider a record type  $R_1$  containing a field  $f_1 : R_2$ , where the record type  $R_2$  contains a field  $f_2 : R_3$ , and the record type  $R_3$  contains a boolean field  $f_3$ :

```
record R1 : Set where
  field f1 : R2
```

```
record R2 : Set where
  field f2 : R3
```

```
record R3 : Set where
  field f3 : Bool
```

(Here Set is a type of small types.) Given a value  $x : R_1$ , how do you invert the boolean value contained within it? Manual code can easily become somewhat awkward:

```
record x { f1 = record (R1.f1 x) { f2 =
  record (R2.f2 (R1.f1 x))
  { f3 = not (R3.f3 (R2.f2 (R1.f1 x))) } } }
```

(The notation  $R_1.f_1 x$  stands for the projection of the field  $R_1.f_1$  from the record value  $x$ .) Consider also the additional complication if the field  $f_3$  is a finite map from strings to Bool, and you want to update the boolean that the string “f<sub>0</sub>o” maps to (if any).

With lenses these updates are easy. Assume that three lenses are given,  $r_1$  corresponding to  $f_1$ ,  $r_2$  corresponding to  $f_2$ , and  $r_3$  corresponding to  $f_3$ :

```
r1 : Lens R1 R2
r2 : Lens R2 R3
r3 : Lens R3 Bool
```

Then one can perform the inversion of the boolean value contained in  $x$  by composing the three lenses:

```
modify (r3 ∘ r2 ∘ r1) not x
```

The question addressed in this paper is whether very well-behaved lenses work equally well for *dependent* records, where the types of later fields can depend on the values of previous fields, like the following type:

```
record Q : Set where
  field
    dividend : ℤ
    divisor   : ℕ
    non-zero  : ¬ divisor ≡ 0
```

The answer, in short:

- No matter how the type family of very well-behaved dependent lenses is defined, if it satisfies a certain, natural property, then it is *impossible* to define a general composition operator for these lenses (see Section 5).
- However, one can define a family of types  $DLens S V$  of dependent lenses, where  $V$  depends on  $S$ , in such a way that getters and setters, and functions like modify, can be defined (Section 6).
- Furthermore it is possible to define a notion of composition that seems to work well in practice (Section 6.4).

These results have been formalised using Agda. This document is a literate Agda file, every piece of code—except for inline code in text paragraphs—has been type-checked. (I have hidden some pieces of code, and changed the order in which the code is displayed.) There is also other, supporting formalisation for several statements mentioned in the text.

The formalisation has been developed with the K rule turned off. Equality of functions is assumed to be extensional, and I assume that a propositional truncation operator (The Univalent Foundations Program 2013) is available. Some statements have been proved under the assumption of univalence (The Univalent Foundations Program 2013), and other statements under the assumption of uniqueness of identity proofs for Set. These two principles are not consistent with each other, so in the text I point out when I make use of them. In the formalisation I track their use using types.

## 2. Non-dependent lenses

Very well-behaved *non-dependent* lenses from a source type  $S$  to a view type  $V$  are often specified to consist of two functions, a getter and a setter, satisfying three laws. In Agda we can phrase this as a dependent record:

```
record Lens (S V : Set) : Set where
  field
    get      : S → V
    set      : S → V → S
    get-set  : ∀ s v → get (set s v) ≡ v
    set-get  : ∀ s → set s (get s) ≡ s
    set-set  : ∀ s v1 v2 → set (set s v1) v2 ≡ set s v2
```

The getter provides a view of the source value, and the setter modifies a source value with a possibly different view. The get-set law states that set really updates the view, the set-get law states that if you update the view with the original view, then nothing happens, and the set-set law states that if you update the view twice (without intervening changes), then the first update is ignored.

Lenses defined in this way form a monoid. Defining an identity lens is easy (the proofs are omitted):

```
id : {S : Set} → Lens S S
id = record
  { get = λ x → x
  ; set = λ _ s → s
  }
```

(The notation  $\{S : \text{Set}\} \rightarrow \dots$  means that  $S$  is an *implicit* argument, that does not need to be given explicitly if Agda can infer it.) We can also define composition of lenses:

```
◦_ : {S V1 V2 : Set} →
  Lens V1 V2 → Lens S V1 → Lens S V2
l1 ◦ l2 = record
  { get = λ s → get l1 (get l2 s)
```

```
; set = λ s v2 →
  let v1 = set l1 (get l2 s) v2 in
  set l2 s v1
}
```

(Here I write  $\text{get } l_1 s$  rather than  $\text{Lens.get } l_1 s$ . Further down I also write  $\text{get } s$ , omitting the lens argument.) One can also prove that composition is associative and that  $\text{id}$  is a left and right unit for composition (even in the absence of the K rule), but the proofs are omitted here.

## 3. No first projection lens

Let us now consider whether there can be a non-dependent lens corresponding to the first projection from a  $\Sigma$ -type (a pair type where the type of the second component can depend on the value of the first component).

It is not so hard to see that this is impossible. Consider the following  $\Sigma$ -type, consisting of a boolean and a proof requiring this boolean to be true:

```
Unit : Set
Unit = Σ b : Bool · b ≡ true
```

There is exactly one value in this type (people familiar with homotopy type theory may recognise it as a singleton type):

```
unit : Unit
unit = (true , refl)
```

```
all-values-equal : (u1 u2 : Unit) → u1 ≡ u2
all-values-equal (true , refl) (true , refl) = refl
```

(Here  $\text{refl}$  is the equality type's constructor.)

Assume for a contradiction that a first projection lens exists for the Unit type:

```
first : Lens Unit Bool
```

By using the get-set law twice and all-values-equal once we get a contradiction:<sup>1</sup>

```
true ≡⟨ by (get-set first) ⟩
get first (set first unit true) ≡⟨ by all-values-equal ⟩
get first (set first unit false) ≡⟨ by (get-set first) ⟩
false ■
```

Note that this contradiction could be proved without assuming that  $\text{get } first p$  is the first projection of the pair  $p$ . Furthermore the proof used only one of the lens laws, get-set.

## 4. Alternative definitions

Before moving on to dependent lenses, let us consider some alternative definitions of very well-behaved non-dependent lenses.

<sup>1</sup>Here  $\text{by}$  is a tactic that, given a lemma, tries to prove an equality.

Pierce and Schmitt (2003) note that for very well-behaved lenses as defined above (but in set theory), satisfying the extra assumption that the range of the set function is the entire source set, the source set is in bijective correspondence with the cartesian product of the view set and some other set. It has been suggested<sup>2</sup> that one can define the type of very well-behaved lenses as a dependent pair ( $\Sigma$ -type) consisting of a type and a bijection (denoted with a double arrow below):

$$\begin{aligned} \text{Bijection-lens} & : \text{Set} \rightarrow \text{Set} \rightarrow \text{Set}_1 \\ \text{Bijection-lens } S V & = \Sigma R : \text{Set} \cdot S \leftrightarrow R \times V \end{aligned}$$

However, this type is not in general isomorphic to the one defined above:  $\text{Lens } \perp \perp$  (where  $\perp$  is the empty type) is isomorphic to the unit type, but  $\text{Bijection-lens } \perp \perp$  is isomorphic to  $\text{Set}$ .

Capriotti (2014) has introduced *higher lenses* (if you do not understand this definition, do not worry about it, it is not essential to understand the rest of the text):

$$\begin{aligned} \text{Higher-lens} & : \text{Set} \rightarrow \text{Set} \rightarrow \text{Set}_1 \\ \text{Higher-lens } S V & = \\ & \Sigma \text{get} : (S \rightarrow V) \cdot \\ & \Sigma P : (\| V \| \rightarrow \text{Set}) \cdot \\ & (\lambda v \rightarrow \Sigma s \cdot \text{get } s \equiv v) \equiv (\lambda v \rightarrow P | v |) \end{aligned}$$

Here  $\| V \|$  stands for the *propositional truncation* of  $V$ . This type is isomorphic to the set<sup>3</sup> quotient of  $V$  with the trivial relation; the constructor  $|\_| : V \rightarrow \| V \|$  shows that  $\| V \|$  is inhabited if  $V$  is. When the source type is a set the type of higher lenses is isomorphic to the type of non-dependent lenses given in Section 2, assuming univalence. (Capriotti (2014) did not present a complete proof. For a formal proof developed by Capriotti, Andrea Vezzosi and myself, see this paper’s accompanying code.)

In the presence of univalence these higher lenses are isomorphic to the following definition of “equivalence-lenses” (found by Andrea Vezzosi and myself), which are similar to the bijection-lenses mentioned above:

$$\begin{aligned} \text{ELens} & : \text{Set} \rightarrow \text{Set} \rightarrow \text{Set}_1 \\ \text{ELens } S V & = \Sigma R : \text{Set} \cdot (S \simeq (R \times V)) \times (R \rightarrow \| V \|) \end{aligned}$$

Here  $A \simeq B$  is the type of *equivalences* between the types  $A$  and  $B$ . When  $A$  and  $B$  are sets this concept from homotopy type theory is in bijective correspondence with the type of bijections between  $A$  and  $B$ . The extra component requiring the view type to be (“merely”) inhabited whenever the remainder type  $R$  is inhabited ensures that, in the presence of univalence,  $\text{ELens } \perp \perp$  is isomorphic to the unit type, so this type can be seen as a more well-behaved variant of  $\text{Bijection-lens}$ . More generally, still in the presence of univalence,  $\text{ELens } S V$  is isomorphic to  $\text{Higher-lens } S V$ , and, if

<sup>2</sup>I do not know who first suggested this.

<sup>3</sup>A concept from homotopy type theory. A type is a set if and only if it satisfies the principle of uniqueness of identity proofs, see Section 6.3.

$S$  is a set, to  $\text{Lens } S V$ . The definition of dependent lenses in Section 6 will be based on this type.

Finally I present an isomorphic variant of  $\text{Equivalence-lens}$ , that may provide a somewhat different perspective on the definition:

$$\begin{aligned} \text{Equivalence-lens}' & : \text{Set} \rightarrow \text{Set} \rightarrow \text{Set}_1 \\ \text{Equivalence-lens}' S V & = \\ & \Sigma \text{get} : (S \rightarrow V) \cdot \\ & \Sigma R : \text{Set} \cdot \\ & \Sigma \text{remainder} : (S \rightarrow R) \cdot \\ & \text{Is-equivalence } (\lambda s \rightarrow \text{remainder } s, \text{get } s) \times \\ & \text{Surjective remainder} \end{aligned}$$

Here  $\text{Is-equivalence } f$  means that  $f$  is an equivalence; this holds if and only if  $f$  has a left and right inverse. The type states that a lens is a getter function, a remainder type, and a *surjective* function from the source type to the remainder type, satisfying the property that the given combination of the remainder function and the getter is an equivalence.

## 5. Composition is impossible

Let us now consider *dependent* lenses. In this section I will show that, given certain assumptions, it is impossible to define composition for such lenses.

I assume that the type of dependent lenses has the following type signature:

$$\text{DLens} : (S : \text{Set}) (V : S \rightarrow \text{Set}) \rightarrow \text{Set}$$

Note that the view type has been replaced by a family of view types, dependent on the source type. I also assume that there is a (dependently typed) getter function:

$$\text{dget} : \forall \{S V\} \rightarrow \text{DLens } S V \rightarrow (s : S) \rightarrow V s$$

Finally I assume that *non-dependent* dependent lenses are not very different from regular non-dependent lenses, by requiring that there is a retraction (split surjection) from the former to the latter, relating  $\text{dget}$  and  $\text{get}$  in the following way:

$$\begin{aligned} \text{to} & : \forall \{S V\} \rightarrow \text{DLens } S (\lambda \_ \rightarrow V) \rightarrow \text{Lens } S V \\ \text{from} & : \forall \{S V\} \rightarrow \text{Lens } S V \rightarrow \text{DLens } S (\lambda \_ \rightarrow V) \\ \text{to-from} & : \forall \{S V\} (l : \text{Lens } S V) \rightarrow \text{to } (\text{from } l) \equiv l \\ \text{get-to} & : \forall \{S V\} (l : \text{DLens } S (\lambda \_ \rightarrow V)) (s : S) \rightarrow \\ & \text{get } (\text{to } l) s \equiv \text{dget } l s \end{aligned}$$

Here is the partial specification of composition that I am using, parametrised by some source and view types:

$$\begin{aligned} \text{record } & \text{Specification-of-composition} \\ & (S : \text{Set}) (V_1 : S \rightarrow \text{Set}) \\ & (V_2 : (s : S) \rightarrow V_1 s \rightarrow \text{Set}) : \text{Set } \mathbf{where} \end{aligned}$$

$$\begin{aligned} \text{field} & \\ \text{dcomp} & : (l_1 : \text{DLens } S V_1) \\ & (l_2 : \forall s \rightarrow \text{DLens } (V_1 s) (V_2 s)) \rightarrow \end{aligned}$$

$$\begin{aligned} & \text{DLens } S (\lambda s \rightarrow V_2 s (\text{dget } l_1 s)) \\ \text{dget-dcomp} & : \forall l_1 l_2 s \rightarrow \text{dget } (\text{dcomp } l_1 l_2) s \equiv \\ & \text{dget } (l_2 s) (\text{dget } l_1 s) \end{aligned}$$

The composition operator takes a lens and a family of lenses to a lens, and the getter for the resulting lens must be expressible in terms of the getters for the argument lenses.

I will prove that Specification-of-composition is uninhabited when all the type parameters are instantiated with (constant functions returning) Bool:

$$\begin{aligned} \text{no-composition-operator} & : \\ & \neg \text{Specification-of-composition Bool } (\lambda \_ \rightarrow \text{Bool}) \\ & \quad (\lambda \_ \_ \rightarrow \text{Bool}) \end{aligned}$$

Assume for a contradiction that this instance of the specification is inhabited, giving us access to the dcomp function and the dget-dcomp law. The existence of the retraction mentioned above ensures that we can define a corresponding composition operator for non-dependent lenses:

$$\begin{aligned} \text{comp} & : \text{Lens Bool Bool} \rightarrow (\text{Bool} \rightarrow \text{Lens Bool Bool}) \rightarrow \\ & \quad \text{Lens Bool Bool} \\ \text{comp } l_1 l_2 & = \text{to } (\text{dcomp } (\text{from } l_1) (\lambda b \rightarrow \text{from } (l_2 b))) \end{aligned}$$

A calculation shows that this operator satisfies a law corresponding to dget-dcomp, see Figure 1.

The comp function will be used together with the following family of lenses, that is the identity for true and swap for false:

$$\begin{aligned} \text{id-or-swap} & : \text{Bool} \rightarrow \text{Lens Bool Bool} \\ \text{id-or-swap true} & = \text{id} \\ \text{id-or-swap false} & = \text{swap} \end{aligned}$$

Here swap is defined in the following way (proofs omitted):

$$\begin{aligned} \text{swap} & : \text{Lens Bool Bool} \\ \text{swap} & = \mathbf{record} \\ & \{ \text{get} = \text{not} \\ & \quad ; \text{set} = \lambda \_ v \rightarrow \text{not } v \\ & \} \end{aligned}$$

A simple case analysis shows that id-or-swap satisfies the following property:

$$\begin{aligned} \text{get-id-or-swap} & : \forall b \rightarrow \text{get } (\text{id-or-swap } b) b \equiv \text{true} \\ \text{get-id-or-swap true} & = \text{refl} \\ \text{get-id-or-swap false} & = \text{refl} \end{aligned}$$

Now we can use composition to construct the following strange lens:

$$\begin{aligned} \text{strange} & : \text{Lens Bool Bool} \\ \text{strange} & = \text{comp id id-or-swap} \end{aligned}$$

This lens is strange because its getter is constant:

$$\begin{aligned} \text{get-strange} & : \forall b \rightarrow \text{get strange } b \equiv \text{true} \\ \text{get-strange } b & = \\ \text{get strange } b & \quad \equiv \langle \rangle \\ \text{get } (\text{comp id id-or-swap } b) & \equiv \langle \text{ by get-comp } \rangle \\ \text{get } (\text{id-or-swap } b) & \\ \quad (\text{get id } b) & \quad \equiv \langle \rangle \\ \text{get } (\text{id-or-swap } b) b & \equiv \langle \text{ by } (\text{get-id-or-swap } b) \rangle \\ \text{true} & \quad \blacksquare \end{aligned}$$

In combination with the get-set law this leads to a contradiction:

$$\begin{aligned} \text{true} & \quad \equiv \langle \text{ by get-strange } \rangle \\ \text{get strange} & \\ \quad (\text{set strange true false}) & \equiv \langle \text{ by } (\text{get-set strange}) \rangle \\ \text{false} & \quad \blacksquare \end{aligned}$$

Note that the proof above only uses one lens law, get-set (just like the proof in Section 3). Dropping the set-get or set-set laws from the definition of non-dependent lenses would not help.

One can perhaps object that the specification of composition is too strong: why should the family of lenses  $l_2$  be allowed to inspect its argument? In Section 6.4 a variant of composition is defined. This notion restricts the family of lenses by not giving it access to values from the source type, but only from the remainder type.

## 6. Dependent lenses

Notwithstanding the negative result in the previous section, this section contains a definition of dependent lenses. The definition is based on the equivalence-lenses from Section 4. These equivalence-lenses contain an equivalence between the source type  $S$  and the cartesian product  $R \times V$  of the remainder type  $R$  and the view type  $V$ , indicating that  $V$  is somehow a part of  $S$ .

For dependent lenses the view type is replaced by a family of types, so one tentative idea may be to replace the cartesian product with a  $\Sigma$ -type  $\Sigma r : R \cdot V r$ . This does not work, because  $V$  does not depend on  $R$ , but on  $S$ . However, if  $V$  is seen as a part of  $S$ , then it might seem reasonable that it should not in general depend on “all” of  $S$ , but only on the remainder type,  $R$ . The following definition therefore introduces an alternative view family  $V'$  that depends on  $R$ , and is restricted to be a variant of  $V$ :

$$\mathbf{record} \text{DLens } (S : \text{Set}) (V : S \rightarrow \text{Set}) : \text{Set}_1 \mathbf{where}$$

$$\begin{aligned} \mathbf{field} \\ R & : \text{Set} \\ V' & : R \rightarrow \text{Set} \\ \text{equiv} & : S \simeq (\Sigma r : R \cdot V' r) \\ \text{inhabited} & : \forall r \rightarrow \parallel V' r \parallel \end{aligned}$$

$$\begin{aligned} \text{remainder} & : S \rightarrow R \\ \text{remainder } s & = \text{fst } (\text{to equiv } s) \end{aligned}$$

```

get-comp :  $\forall l_1 l_2 b \rightarrow \text{get } (\text{comp } l_1 l_2) b \equiv \text{get } (l_2 b) (\text{get } l_1 b)$ 
get-comp  $l_1 l_2 b =$ 
  get (comp  $l_1 l_2$ )  $b$   $\equiv\langle$ 
  get (to (dcomp (from  $l_1$ ) ( $\lambda b \rightarrow$  from ( $l_2 b$ ))))  $b$   $\equiv\langle$  by get-to  $\rangle$ 
  dget (dcomp (from  $l_1$ ) ( $\lambda b \rightarrow$  from ( $l_2 b$ )))  $b$   $\equiv\langle$  by dget-dcomp  $\rangle$ 
  dget (from ( $l_2 b$ )) (dget (from  $l_1$ )  $b$ )  $\equiv\langle$  by get-to  $\rangle$ 
  get (to (from ( $l_2 b$ ))) (dget (from  $l_1$ )  $b$ )  $\equiv\langle$  by get-to  $\rangle$ 
  get (to (from ( $l_2 b$ ))) (get (to (from  $l_1$ ))  $b$ )  $\equiv\langle$  by to-from  $\rangle$ 
  get ( $l_2 b$ ) (get  $l_1 b$ )  $\blacksquare$ 

```

**Figure 1.** The get-comp law. Note that if the explanation has been omitted in an equational reasoning step, as in  $\dots \equiv\langle \dots$ , then the two sides are equal by definition.

### field

variant :  $\forall \{s\} \rightarrow V' (\text{remainder } s) \equiv V s$

Here to equiv is the forward component of the equivalence. For every source value  $s$  we can compute a remainder value  $s$ , and the last field requires  $V' (\text{remainder } s)$  to be equal to  $V s$ .

### 6.1 Getters and setters

Given a dependent lens we can define a getter and a setter:

```

get' : (s : S) → V' (remainder s)
get' s = snd (to equiv s)

```

```

get : (s : S) → V s
get s = cast variant (get' s)

```

```

set' : (s : S) → V' (remainder s) → S
set' s v' = from equiv (remainder s, v')

```

```

set : (s : S) → V s → S
set s v = set' s (cast (sym variant) v)

```

These definitions make use of the cast function to “transport” a value from one type to another via an equality. The cast function is in turn defined using subst:

```

subst :  $\forall \{a p\} \{A : \text{Set } a\} (P : A \rightarrow \text{Set } p) \rightarrow$ 
   $\forall \{x y : A\} \rightarrow x \equiv y \rightarrow P x \rightarrow P y$ 
subst _ refl  $p = p$ 

```

```

cast : {A B : Set} → A  $\equiv$  B → A → B
cast = subst id

```

The last definition also uses symmetry:

```

sym :  $\forall \{a\} \{A : \text{Set } a\} \{x y : A\} \rightarrow x \equiv y \rightarrow y \equiv x$ 
sym refl = refl

```

(Some functions above have been given universe-polymorphic types, because they are used with “large” arguments.)

We can also define a function corresponding to the modify function used in the introduction:

```

modify : (s : S) → (V s → V s) → S
modify s f = set s (f (get s))

```

### 6.2 Lens laws

It is possible to prove variants of the usual lens laws. The set-get law can be established using simple equational reasoning, see Figure 2. This proof uses a lemma, cast-sym-cast, which states that if you cast in one direction, and then in the opposite direction, then you get back what you started with:

```

cast-sym-cast : {A B : Set} (eq : A  $\equiv$  B) {x : A} →
  cast (sym eq) (cast eq x)  $\equiv$  x
cast-sym-cast refl = refl

```

The following variant of cast-sym-cast will be used below:

```

cast-cast-sym : {A B : Set} (eq : A  $\equiv$  B) {x : B} →
  cast eq (cast (sym eq) x)  $\equiv$  x
cast-cast-sym refl = refl

```

The other two non-dependent lens laws are not well-typed in this setting. For instance,  $\text{get}' (\text{set}' s v')$  has type  $V' (\text{remainder } (\text{set}' s v'))$ , not  $V' (\text{remainder } s)$ . However, we can get around this problem by observing that set and set' do not change the remainder:

```

remainder-set' :  $\forall s v' \rightarrow$ 
  remainder (set' s v')  $\equiv$  remainder s

```

```

remainder-set' s v' =
  remainder (set' s v')  $\equiv\langle$ 
  fst (to equiv (from equiv s'))  $\equiv\langle$  cong fst (to-from equiv _)  $\rangle$ 
  fst s'  $\equiv\langle$ 
  remainder s  $\blacksquare$ 

```

**where**

```

set'-get' : ∀ s → set' s (get' s) ≡ s
set'-get' s =
  set' s (get' s)                ≡⟨ ⟩
  from equiv (remainder s , snd (to equiv s)) ≡⟨ ⟩
  from equiv (fst (to equiv s) , snd (to equiv s)) ≡⟨ ⟩
  from equiv (to equiv s)       ≡⟨ by (from-to equiv) ⟩
  s                               ■

set-get : ∀ s → set s (get s) ≡ s
set-get s =
  set s (get s)                ≡⟨ ⟩
  set' s (cast (sym variant) (cast variant (get' s))) ≡⟨ by (cast-sym-cast variant) ⟩
  set' s (get' s)              ≡⟨ by set'-get' ⟩
  s                               ■

```

**Figure 2.** The set'-get' and set-get laws.

```

s' = (remainder s , v')

remainder-set : ∀ s v → remainder (set s v) ≡ remainder s
remainder-set s v =
  remainder (set s v)          ≡⟨ ⟩
  remainder
    (set' s (cast (sym variant) v)) ≡⟨ remainder-set' _ _ ⟩
  remainder s                   ■

```

These proofs are not proved using the by tactic, because later results depend on the structure of the proofs. Instead I have used the cong lemma, which expresses the fact that every non-dependent function respects equality:

```

cong : ∀ {a b} {A : Set a} {B : Set b} →
  (f : A → B) {x y : A} → x ≡ y → f x ≡ f y
cong _ refl = refl

```

The get'-set' law can now be stated by using the subst function to transport from one type to another:

```

get'-set' :
  ∀ s v' →
  subst V' (remainder-set' s v') (get' (set' s v')) ≡ v'

```

This statement of the get'-set' law can be proved by using equational reasoning, see Figure 3. The proof makes use of two general lemmas. The first one relates subst and cong:

```

subst-◦ :
  ∀ {a b p} {A : Set a} {B : Set b} {x y : A}
  {P : B → Set p} {f : A → B} {p : P (f x)}
  (eq : x ≡ y) →
  subst (P ◦ f) eq p ≡ subst P (cong f eq) p
subst-◦ refl = refl

```

The second one makes it possible to remove an instance of subst (the name of this lemma is taken from The Univalent Foundations Program (2013)):

```

apd : {A : Set} {B : A → Set} {x y : A}
  (f : (x : A) → B x) (eq : x ≡ y) →
  subst B eq (f x) ≡ f y
apd _ refl = refl

```

The V-set lemma states that the view type stays unchanged when the source parameter is changed using set:

```

V-set : ∀ s v → V (set s v) ≡ V s
V-set s v =
  V (set s v)                ≡⟨ sym variant ⟩
  V' (remainder (set s v)) ≡⟨ cong V' (remainder-set s v) ⟩
  V' (remainder s)          ≡⟨ variant ⟩
  V s                         ■

```

This lemma can be used to state the get-set law, see Figure 4. Here I have relegated some steps of equational reasoning to an omitted lemma.

The set'-set' law can be stated by using remainder-set' to “correct” the type of the second view argument, see Figure 5. The law is proved by using  $\Sigma \equiv$ , which gives a general way to prove that two values from a  $\Sigma$ -type are equal:

```

Σ≡ :
  {A : Set} {B : A → Set} {x y : Σ x : A · B x} →
  (eq : fst x ≡ fst y) →
  subst B eq (snd x) ≡ snd y →
  x ≡ y
Σ≡ refl refl = refl

```

Finally we can state and prove the set-set law, see Figure 6. Here I have again relegated some steps of equational reasoning to an omitted lemma.

```

get'-set' s v' =
  subst V' (remainder-set' s v') (get' (set' s v'))    ≡⟨ ⟩
  subst V' (cong fst eq) (get' (set' s v'))           ≡⟨ by (subst-◦ eq) ⟩
  subst (V' ◦ fst) eq (get' (set' s v'))             ≡⟨ ⟩
  subst (V' ◦ fst) eq (snd (to equiv (from equiv s'))) ≡⟨ by (apd snd eq) ⟩
  snd s'                                             ≡⟨ ⟩
  v'                                                 ■
where
s' = (remainder s , v')
eq = to-from equiv s'

```

**Figure 3.** The get'-set' law.

```

get-set : ∀ s v → cast (V-set s v) (get (set s v)) ≡ v
get-set s v =
  cast (V-set s v) (get (set s v))                    ≡⟨ ⟩
  cast (V-set s v) (cast variant (get' (set s v)))    ≡⟨ by lemma ⟩
  cast variant (cast (cong V' (remainder-set s v)) (get' (set s v))) ≡⟨ by (subst-◦ (remainder-set s v)) ⟩
  cast variant (subst V' (remainder-set s v) (get' (set s v)))    ≡⟨ ⟩
  cast variant (subst V' (remainder-set s v) (get' (set' s (cast (sym variant) v)))) ≡⟨ by get'-set' ⟩
  cast variant (cast (sym variant) v)                 ≡⟨ by (cast-cast-sym variant) ⟩
  v                                                    ■

```

**Figure 4.** The get-set law.

```

set'-set' : ∀ s v1' v2' → set' (set' s v1') v2' ≡ set' s (subst V' (remainder-set' s v1') v2')
set'-set' s v1' v2' =
  set' (set' s v1') v2'                               ≡⟨ ⟩
  from equiv (remainder (set' s v1'), v2')           ≡⟨ by (Σ-≡ (remainder-set' s v1') refl) ⟩
  from equiv (remainder s , subst V' (remainder-set' s v1') v2') ≡⟨ ⟩
  set' s (subst V' (remainder-set' s v1') v2')       ■

```

**Figure 5.** The set'-set' law.

### 6.3 Relation to non-dependent lenses

The principle of uniqueness of identity proofs (UIP) for a given type  $A$  states that, for any two values of type  $A$ , all equality proofs between those two values are equal:

```

Uniqueness-of-identity-proofs :
  ∀ {a} → Set a → Set a
Uniqueness-of-identity-proofs A =
  {x y : A} → (p q : x ≡ y) → p ≡ q

```

If we assume that this property holds for  $\text{Set}$ , the type of small types, then non-dependent dependent lenses are isomorphic to equivalence-lenses:

```

Uniqueness-of-identity-proofs Set →
  {S V : Set} →
  Σ eq : DLens S (λ _ → V) ≃ ELens S V ·
  ∀ l s → get l s ≡ eget (to eq l) s

```

(Here  $\text{eget}$  is the  $\text{get}$  function for equivalence-lenses.) Note that the assumption that  $\text{Set}$  satisfies UIP is not compatible with univalence (but it holds in Agda if the  $\text{K}$  rule is turned on). I have not managed to determine if non-dependent dependent lenses are isomorphic to equivalence-lenses in the presence of univalence.

The argument in Section 5 showing that composition is impossible for dependent lenses was carried out using regular lenses,  $\text{Lens}$ . It was also restricted to dependent lenses living

$$\begin{aligned}
& \text{set-set} : \forall s v_1 v_2 \rightarrow \text{set} (\text{set } s v_1) v_2 \equiv \text{set } s (\text{cast } (V\text{-set } s v_1) v_2) \\
& \text{set-set } s v_1 v_2 = \\
& \quad \text{set} (\text{set } s v_1) v_2 \quad \equiv \langle \rangle \\
& \quad \text{set}' (\text{set}' s (\text{cast } (\text{sym variant}) v_1)) (\text{cast } (\text{sym variant}) v_2) \quad \equiv \langle \text{ by set}'\text{-set}' \rangle \\
& \quad \text{set}' s (\text{subst } V' (\text{remainder-set}' s (\text{cast } (\text{sym variant}) v_1)) \\
& \quad \quad (\text{cast } (\text{sym variant}) v_2)) \quad \equiv \langle \rangle \\
& \quad \text{set}' s (\text{subst } V' (\text{remainder-set } s v_1) (\text{cast } (\text{sym variant}) v_2)) \quad \equiv \langle \text{ by (subst-}\circ (\text{remainder-set } s v_1)) \rangle \\
& \quad \text{set}' s (\text{cast } (\text{cong } V' (\text{remainder-set } s v_1)) (\text{cast } (\text{sym variant}) v_2)) \quad \equiv \langle \text{ lemma } \rangle \\
& \quad \text{set}' s (\text{cast } (\text{sym variant}) (\text{cast } (V\text{-set } s v_1) v_2)) \quad \equiv \langle \rangle \\
& \quad \text{set } s (\text{cast } (V\text{-set } s v_1) v_2) \quad \blacksquare
\end{aligned}$$

**Figure 6.** The set-set law.

in  $\text{Set}$ , but the dependent lenses defined in this section live in  $\text{Set}_1$ . However, the argument works also for equivalence-lenses and larger dependent lenses, even in the absence of univalence. Thus, in the presence of UIP for  $\text{Set}$ , the strong form of composition discussed in Section 5 cannot be defined for  $\text{DLens}$ .

It is also possible to show that, in the presence of UIP for set, the first projection lens cannot always be defined. Let us assume that UIP holds:

$$\text{uip} : \text{Uniqueness-of-identity-proofs } \text{Set}$$

In this case casts between a type and itself can be removed:

$$\begin{aligned}
& \text{drop-cast} : \{A : \text{Set}\} (eq : A \equiv A) (x : A) \rightarrow \\
& \quad \text{cast } eq x \equiv x \\
& \text{drop-cast } eq x = \\
& \quad \text{cast } eq x \quad \equiv \langle \text{ by uip } \rangle \\
& \quad \text{cast refl } x \quad \equiv \langle \rangle \\
& \quad x \quad \blacksquare
\end{aligned}$$

Let us now, as in Section 3, assume that there is a first projection lens for the Unit type:

$$\text{first} : \text{DLens } \text{Unit} (\lambda \_ \rightarrow \text{Bool})$$

We again get a contradiction, see Figure 7. I do not know if a corresponding result can be proved in the presence of univalence.

## 6.4 Composition

If composition cannot be defined, what is the point of dependent lenses? Fortunately it is possible to define a weaker variant of composition. As mentioned in Section 5 the notion of composition defined here does not allow the family of lenses to inspect values from the source type, but only from the remainder type. Composition is defined in Figure 8. The remainder type combines the two remainder types, and the variant of the view type family is defined in terms of the lens family's variant type families. The equivalence is defined using the argument equivalences, using the following associativity-like property for  $\Sigma$ -types:

$$\begin{aligned}
& \Sigma\text{-assoc} : \\
& \quad \{A : \text{Set}\} \{B : A \rightarrow \text{Set}\} \\
& \quad \{C : \Sigma x : A \cdot B x \rightarrow \text{Set}\} \rightarrow \\
& \quad (\Sigma x : A \cdot \Sigma y : B x \cdot C (x, y)) \simeq \\
& \quad (\Sigma p : (\Sigma x : A \cdot B x) \cdot C p)
\end{aligned}$$

Finally the inhabitation and variant proofs are defined using the lens family's inhabitation and variant proofs.

The following minor variant, in which the lens family takes an implicit argument instead of an explicit one, will be used in the examples below:

$$\begin{aligned}
& \text{--}\dot{\_} : \\
& \quad \{S : \text{Set}\} \{V_1 V_2 : S \rightarrow \text{Set}\} \rightarrow \\
& \quad (l_1 : \text{DLens } S V_1) \rightarrow \\
& \quad (\forall \{r\} \rightarrow \\
& \quad \quad \text{DLens } (V' l_1 r) \\
& \quad \quad (\lambda v' \rightarrow V_2 (\text{from } (\text{equiv } l_1) (r, v')))) \rightarrow \\
& \quad \text{DLens } S V_2 \\
& \quad l_1 ; l_2 = \text{dcomp } l_1 (\lambda \_ \rightarrow l_2)
\end{aligned}$$

## 6.5 Constructing lenses

When constructing a lens one can often avoid having to construct all the fields above, in many cases it suffices to give an equivalence:

$$\begin{aligned}
& \text{equivalence-to-lens} : \\
& \quad \{S R : \text{Set}\} \{V : R \rightarrow \text{Set}\} \\
& \quad (eq : S \simeq (\Sigma r : R \cdot V r)) \rightarrow \\
& \quad \text{DLens } S (\lambda s \rightarrow V (\text{fst } (\text{to } eq s))) \\
& \text{equivalence-to-lens } \{S\} \{R\} \{V\} eq = \text{record} \\
& \quad \{ R \quad = \Sigma r : R \cdot \| V r \| \\
& \quad ; V' \quad = \lambda r \rightarrow V (\text{fst } r) \\
& \quad ; \text{equiv} = \\
& \quad \quad S \quad \quad \quad \simeq \langle eq \rangle \\
& \quad (\Sigma r : R \cdot V r) \quad \simeq \langle \Sigma\text{-cong } \simeq \times \| \| \rangle \\
& \quad (\Sigma r : R \cdot \| V r \| \times V r) \quad \simeq \langle \Sigma\text{-assoc} \rangle \\
& \quad (\Sigma p : (\Sigma r : R \cdot \| V r \|) \cdot V (\text{fst } p)) \quad \square
\end{aligned}$$



true	$\equiv\langle$ by (get-set <i>first</i> ) $\rangle$
cast (V-set <i>first</i> unit true) (get <i>first</i> (set <i>first</i> unit true))	$\equiv\langle$ by (drop-cast (V-set <i>first</i> unit true)) $\rangle$
get <i>first</i> (set <i>first</i> unit true)	$\equiv\langle$ by all-values-equal $\rangle$
get <i>first</i> (set <i>first</i> unit false)	$\equiv\langle$ by (drop-cast (V-set <i>first</i> unit false)) $\rangle$
cast (V-set <i>first</i> unit false) (get <i>first</i> (set <i>first</i> unit false))	$\equiv\langle$ by (get-set <i>first</i> ) $\rangle$
false	■

**Figure 7.** A contradiction.

```

dcomp : {S : Set} {V1 V2 : S → Set} →
  (l1 : DLens S V1) →
  (∀ r → DLens (V' l1 r) (λ v' → V2 (from (equiv l1) (r, v')))) →
  DLens S V2
dcomp {S} {V1} {V2} l1 l2 = record
  { R      = Σ r1 : R l1 · R (l2 r1)
  ; V'     = λ p → V' (l2 (fst p)) (snd p)
  ; equiv  = S
              (Σ r1 : R l1 · V' l1 r1)
              (Σ r1 : R l1 · Σ r2 : R (l2 r1) · V' (l2 r1) r2)
              (Σ p : (Σ r1 : R l1 · R (l2 r1)) · V' (l2 (fst p)) (snd p)) □
  ; inhabited = λ p → inhabited (l2 (fst p)) (snd p)
  ; variant  = λ {s} →
              let r1 = fst (to (equiv l1) s) in
              V' (l2 r1) (remainder (l2 r1) (snd (to (equiv l1) s)))
              V2 (from (equiv l1) (to (equiv l1) s))
              V2 s
  }

```

**Figure 8.** Composition for dependent lenses.

```

; inhabited = snd
; variant  = refl
}

```

```

S      ≃⟨ T-× ⟩
T × S □

```

The remainder type  $R$  and the variant of the view family  $V'$  are chosen in such a way that it is easy to construct the inhabited and variant proofs. The equivalence  $\text{equiv}$  is constructed using “equivalence-reasoning”. The first step uses the supplied equivalence, and the last step uses  $\Sigma$ -assoc. The second step uses the fact that  $\Sigma$ -types preserve equivalences in their second argument ( $\Sigma$ -cong) and the fact that any type is equivalent to the cartesian product of its propositional truncation and itself:

$$\simeq \times \parallel \parallel : \{A : \text{Set}\} \rightarrow A \simeq (\parallel A \parallel \times A)$$

Using equivalence-to-lens it is easy to construct an identity lens:

```

id : {S : Set} → DLens S (λ _ → S)
id {S} = equivalence-to-lens (

```

The proof makes use of the fact that the unit type is a left identity for the cartesian product ( $T \times$ ).

## 6.6 Examples

Consider the following record types (where  $\text{Fin } n$  is a type containing  $n$  values):

```

record R1 (n : ℕ) : Set where
  field
  f : Fin n → Fin n
  x : Fin n
  lemma : ∀ y → f y ≡ y

```

```

record R2 : Set where
  field
  n : ℕ
  r1 : R1 n

```

We can easily define lenses for the  $\times$ , and lemma fields (implementation omitted):

$$\begin{aligned} \times & : \{n : \mathbb{N}\} \rightarrow \text{DLens } (R_1 \ n) (\lambda \_ \rightarrow \text{Fin } n) \\ \text{lemma} & : \{n : \mathbb{N}\} \rightarrow \\ & \quad \text{DLens } (R_1 \ n) (\lambda r \rightarrow \forall y \rightarrow R_1.f \ r \ y \equiv y) \end{aligned}$$

I do not see much point in trying to define a lens for the  $f$  field in isolation: if the function is changed, then the lemma may have to be modified. However, we can construct a lens that provides a view of both the function and the lemma:

$$\begin{aligned} f & : \{n : \mathbb{N}\} \rightarrow \\ & \quad \text{DLens } (R_1 \ n) \\ & \quad (\lambda \_ \rightarrow \Sigma f : (\text{Fin } n \rightarrow \text{Fin } n) \cdot \forall y \rightarrow f \ y \equiv y) \end{aligned}$$

It is also possible to define a lens for the  $r_1$  field of  $R_2$ :

$$r_1 : \text{DLens } R_2 (\lambda r \rightarrow R_1 (R_2.n \ r))$$

Finally we can compose lenses.

$$\begin{aligned} n_2 & : \text{DLens } R_2 (\lambda r \rightarrow \text{Fin } (R_2.n \ r)) \\ n_2 & = r_1 \ ; \ \times \\ f_2 & : \text{DLens } R_2 (\lambda r \rightarrow \Sigma f : (\text{Fin } (R_2.n \ r) \rightarrow \text{Fin } (R_2.n \ r)) \cdot \\ & \quad \forall y \rightarrow f \ y \equiv y) \\ f_2 & = r_1 \ ; \ f \\ \text{lemma}_2 & : \text{DLens } R_2 (\lambda r \rightarrow \forall y \rightarrow R_1.f \ (R_2.r_1 \ r) \ y \equiv y) \\ \text{lemma}_2 & = r_1 \ ; \ \text{lemma} \end{aligned}$$

Note that, despite the somewhat complicated type of the composition operator, the composed lenses have reasonable types.

## 6.7 Lenses from containers

This section contains a (perhaps not very useful) observation about the relationship between unary containers and dependent lenses.

A unary container (Abbott et al. 2005) consists of a type of shapes  $S$  and for each shape  $s$  a type of positions  $P \ s$ :

```
record Container : Set1 where
  constructor _▷_
  field
    Shape   : Set
    Position : Shape → Set
```

Given such a container one can construct a type constructor:

$$\begin{aligned} \llbracket \_ \rrbracket & : \text{Container} \rightarrow \text{Set} \rightarrow \text{Set} \\ \llbracket C \rrbracket A & = \Sigma s : \text{Shape } C \cdot (\text{Position } C \ s \rightarrow A) \end{aligned}$$

The type  $\llbracket C \rrbracket A$  consists of pairs of a shape and an indexing function mapping this shape's positions to values of type  $A$ . Let us name the projections:

$$\begin{aligned} \text{shape} & : \{C : \text{Container}\} \{A : \text{Set}\} \rightarrow \\ & \quad \llbracket C \rrbracket A \rightarrow \text{Shape } C \\ \text{shape} & = \text{fst} \\ \text{index} & : \{C : \text{Container}\} \{A : \text{Set}\} \\ & \quad (x : \llbracket C \rrbracket A) \rightarrow \text{Position } C (\text{shape } x) \rightarrow A \\ \text{index} & = \text{snd} \end{aligned}$$

A standard example of a unary container is the list type family, which can be represented as a lens in the following way:

```
List : Container
List = ℕ ▷ Fin
```

The shape of a list is its length, and a list of length  $n$  has  $n$  positions. The following container instead gives rise to streams:

```
Stream : Container
Stream = T ▷ λ _ → ℕ
```

There is only one shape, and the type of positions for this shape is the natural numbers. More generally, in a certain extensional type theory unary containers can represent every strictly positive simple type in one variable (Abbott et al. 2005).

It may be interesting to note that a unary container immediately gives rise to a dependent lens:

```
container-to-lens :
  (C : Container) {A : Set} →
  DLens (llbracket C llbracket A) (λ xs → Position C (shape xs) → A)
container-to-lens C {A} = equivalence-to-lens (
  llbracket C llbracket A                                     ≃⟨⟩
  (Σ s : Shape C · (Position C s → A)) □)
```

The source type is the container type, and the view family consists of functions from positions to values. The getter function takes a container object and a matching position, and returns the value at that position:

```
container-get :
  (C : Container) {A : Set} (xs : llbracket C llbracket A) →
  Position C (shape xs) → A
container-get C = get (container-to-lens C)
```

The setter function does not update the value at a given position, it updates the entire position-to-value function:

```
container-set :
  (C : Container) {A : Set} (xs : llbracket C llbracket A) →
  (Position C (shape xs) → A) → llbracket C llbracket A
container-set C = set (container-to-lens C)
```

## 7. Related work

I am not aware of any prior work on lenses with the kind of dependency described above. Ahman and Uustalu (2014) describe dependently typed update lenses. Such a lens is defined for a given directed container, consisting of a set  $S$ , an  $S$ -indexed family of sets  $P$ , and three operations satisfying certain laws, and consists of a set  $X$  of sources and a function  $\text{act} : X \rightarrow \Sigma s : S.Ps \rightarrow X$  satisfying some laws. The  $\text{act}$  function can be decomposed into two functions,  $g : X \rightarrow S$ , and  $s : (x : X) \rightarrow P(g\ x) \rightarrow X$ . This can be compared to a dependent lens  $\text{DLens } X\ S$ , for which the  $\text{get}$  function has type  $(x : X) \rightarrow S\ x$ , and the  $\text{set}$  function has type  $(x : X) \rightarrow S\ x \rightarrow X$ . Perhaps it would be interesting to combine dependent lenses and dependently typed update lenses.

Hofmann et al. (2012) describe symmetric edit lenses, and discuss a lifting that takes a lens between two “modules”  $X$  and  $Y$  to a lens between  $T(X)$  and  $T(Y)$ , where  $T$  is constructed from an arbitrary container of a certain kind.

## 8. Discussion

I have presented a notion of (very well-behaved) dependent lenses, proved that a certain form of composition cannot be defined for such lenses, and provided another form of composition which does work.

Several questions remain unanswered. In Section 6.3 I noticed that, in the presence of UIP for the type of small types, non-dependent lenses are isomorphic to dependent lenses, and the first projection lens cannot be defined for all  $\Sigma$ -types. I do not know whether these properties hold in a setting where UIP is replaced by univalence. Perhaps the given construction does not work very well in this setting.

## Acknowledgements

I would like to thank Paolo Capriotti and Andrea Vezzosi (who are also mentioned in the text above).

The research leading to these results has received funding from the European Research Council under the European Union’s Seventh Framework Programme (FP7/2007-2013) / ERC grant agreement n° 247219. It has also received funding from the Swedish Research Council (621-2013-4879).

## References

- Michael Abbott, Thorsten Altenkirch, and Neil Ghani. Containers: Constructing strictly positive types. *Theoretical Computer Science*, 342:3–27, 2005. doi:10.1016/j.tcs.2005.06.002.
- Danel Ahman and Tarmo Uustalu. Coalgebraic update lenses. In *Proceedings of the 30th Conference on the Mathematical Foundations of Programming Semantics (MFPS XXX)*, volume 308 of *Electronic Notes in Theoretical Computer Science*, pages 25–48, 2014. doi:10.1016/j.entcs.2014.10.003.
- Paolo Capriotti. Higher lenses. Homotopy Type Theory blog post, at the time of writing available at

<https://homotopytypetheory.org/2014/04/29/higher-lenses/>, 2014.

- J. Nathan Foster, Michael B. Greenwald, Jonathan T. Moore, Benjamin C. Pierce, and Alan Schmitt. Combinators for bi-directional tree transformations: A linguistic approach to the view update problem. In *POPL<sup>®</sup> 2005: The 32nd ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages<sup>®</sup>*, pages 233–246, 2005. doi:10.1145/1040305.1040325.
- Martin Hofmann, Benjamin Pierce, and Daniel Wagner. Edit lenses. In *POPL’12, Proceedings of the 39th Annual ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages*, pages 495–508, 2012. doi:10.1145/2103656.2103715.
- Benjamin C. Pierce and Alan Schmitt. Lenses and view update translation. Unpublished, 2003.
- The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. First edition, 2013.